USING VERSION 67.5 OUTPUT2 NEUTRAL FORMAT OPTION FOR FILE TRANSFER ACROSS DISSIMILAR MACHINES

John Brown and Grant Parker
Northrop Corporation's Aircraft Division
El Segundo, California
and
Ted Rose
The MacNeal-Schwendler Corporation
Los Angeles, California

ABSTRACT:

A new option in Version 67.5 provides the ability to write OUTPUT2 files in a machine-independent format. Northrop's Aircraft Division F/A-18 E/F Program has migrated a large portion of its engineering computer use from a large mainframe to a multi-node client/server networked workstation environment. Finite element modeling and analysis software, both proprietary and commercial, have been migrated to the workstation - including MSC/NASTRAN.

Special features in Northrop's mainframe based proprietary pre and post processing code (NCASA) has required its continued use. This mainframe program requires OUTPUT2 data from MSC/NASTRAN. Prior to the introduction of workstations, both NCASA and MSC/NASTRAN were resident on the mainframe, and transfer of files between the programs was not a problem.

In the workstation environment, the binary incompatibility with the mainframe precludes the use of MSC/NASTRAN on the workstation if NCASA post processing capabilities are required. The new OUTPUT2 option in V67.5, which produces a machine-independent compressed ASCII format, allows all MSC/NASTRAN processing to be accomplished on the workstation which results in large cost savings and productivity gains for Northrop.

The implementation of the OUTPUT2 Neutral Format Option, the RCOUT2 MSC supplied FORTRAN conversion program on the mainframe, and use of the MSC/NASTRAN Data Definition Language (NDDL) are presented.

INTRODUCTION:

Early in 1992, the F/A-18 E/F program at Northrop Aircraft Company replaced its mainframe based proprietary computer aided design (CAD) system with a muti-node client/server networked workstation based commercially available package from Electronic Data System (EDS) - Unigraphics. In the following year, the decision was made to move structural analysis from mainframe computers to the same workstations used for CAD. A transition of this type however efficient does cause some pain for the users.

One major problem resulting from this transition to the workstation was the analysis engineers used an in house developed proprietary software package, Northrop Computer Aided Structural Analysis, (NCASA) to perform finite element pre and post processing. Unfortunately at the time of the changeover, this software only ran on the mainframe.

One of the most critical tasks was to allow users to make analyses runs using MSC/NASTRAN on the workstation and pre and post process using NCASA on the mainframe. When the mainframe to workstation transition began, Version 67 of MSC/NASTRAN was the latest version of the software available. Unfortunately, version 67 was limited in the data it could output for meaningful transmission between dissimilar computers. Version 67 would only output matrices by the use of the OUTPUT4 format. In addition to these serious limitations, the data put out by the OUTPUT4 option was single precision and not the double precision actually calculated internally by MSC/NASTRAN. Often this degradation in precision has significant impact on results post processing. Further, to post process MSC/NASTRAN results using NCASA, table data is required from MSC/NASTRAN. There was no easy and convenient way to transmit tables from MSC/NASTRAN from one dissimilar computer to another.

With the release of MSC/NASTRAN Version 67.5, the ability to write OUTPUT2 files which include matrices and tables formatted in compressed ASCII was introduced. This allowed for the first time, the ability to easily transmit tables as well as matrices between dissimilar machines.

DISCUSSION:

The process for transmitting data via the OUTPUT2 compressed ASCII format consisted of first running MSC/NASTRAN on the workstation, outputting the necessary results via DMAP to an OUTPUT2 compressed ASCII file, then transmitting these data to the mainframe.

The file is transmitted to the mainframe by using File Transfer Protocol (FTP). Once on the mainframe, an MSC supplied FORTRAN program (RCOUT2) is executed to convert the compressed ASCII back to mainframe compatible binary format. This binary compatible OUTPUT2 file is then used for NCASA post processing.

Two DMAP's needed to be written for each analysis, i. e., structured solutions such as 101, 103, etc., and unstructured solutions such as solution 24, in order to process the data into an OUTPUT2 compressed ASCII file.

First a subdmap had to be written for all of the structured solutions, 100 series, and a more conventional DMAP for the unstructured solutions. A requirement is that each data block had to be described in the MSC/NASTRAN Data Definition Language(NDDL).

Due to a separate NDDL for both the structured solutions and unstructured solutions, care was exercised in writing the respective DMAP's. Additional care was used in checking the respective NDDL in order to obtain the correct format for outputting matrices or tables.

In order to make the DMAP's easier for the user, and to control the output of the F06 file, the use of (plot) was utilized in the case control. For example, if displacements are desired for post processing, but not needed in hard copy text output, one would simply specify "disp(plot)=all" in the case control.

CONCLUSIONS:

A series of DMAP alters were written for MSC/NASTRAN users at Northrop Corporation. These alters have been written to allow Northrop engineers the ability to process all MSC/NASTRAN jobs on the HP workstation and post process on the mainframe using Northrop's proprietary pre and post program NCASA.

All of the DMAPs utilize the capability in version 67.5 which allows output of both tables and matrices in OUTPUT4 compressed ASCII formatted files, thus allowing the easy transfer of these data between dissimilar machines.

REFERENCES:

[1] The MacNeal-Schwendler Corporation, MSC/NASTRAN Release Notes for Version 67.5

[2] The MacNeal-Schwendler Corporation, MSC/NASTRAN User's Manual, Volumes 1 and 2.

[3] The MacNeal-Schwendler Corporation, MSC/NASTRAN Programmer's Manual, Volumes 1 through 5.

[4] Northrop Corporation, NCASA User's Manual.

APPENDIX A - ALTER for Solution 24

```
$
compile subdmap = sol24, souin = mscsou, nolist, noref $
$
$*********************************************************
$
$        ******* msc_v675_sol24.alt *******
$
$ this dmap runs in msc/nastran solution 24.  it writes the necessary datasets for norpost  processing.
$ basically, nastran version 67.5  is run and the output2 file is written in compressed ascii format.  the
$ results are then transmitted to the ibm via ftp.  next a conversion program is run to convert the
$ compressed ascii  formatted file to ibm binary.  this file is then input to norpost.  the output database
$ from this run can be directly input to ncasa.
$
$    for the nastran run an assign card must be added above the 'id' card.
$
$    assign output2='ncasa.op2', status=new,
$         form=unformatted, unit=12
$
$ if transferring data to IBM mainframe the assign card should set form=formatted for compressed ASCII
$ output as shown below:
$
$    assign output2='ncasa.op2', status=new,
$         form=formatted, unit=12
$
$ in addition, the nastran card should be added to take advantage of the sparse solver plus a param card in
$ the bulk data as follows:
$
$     nastran sparse=25
$
$     param,newseq,6
$
$*********************************************************
$
$ this alter is for solution 24 this alter will print out maximums for spcforces, and displacements and applied
$ loads  similar to solution 101
$
alter 160
$
vecplot   qg,bgpdt,eqexin,cstm,casecc,/qgsum/grdpnt/0/1/
               'spcforce' $
vecplot   qg,bgpdt,eqexin,cstm,casecc,/qgbb//0/5/'maximum'/
               'spcforce'/'s' $
vecplot   ugv,bgpdt,eqexin,cstm,casecc,/ugvbb//0/5/'maximum'/
               'displace'/'ments' $
vecplot   pgg,bgpdt,eqexin,cstm,casecc,/pgbb//0/5/'maximum'/
      'applied'/'loads' $
```

APPENDIX A - ALTER for Solution 24 (Concluded)

```
$
$ ********** end of maximum alter *************
$
alter 176
$
    output2 ,,,,, //-1 / 12 /'statics'/ $
    output2 casecc, , , , // 0 / 12 / $
    output2 cstm, gpl, bgpdt, gpl, gpdt // 0 / 12 / $
    output2 uset, eqexin ,geom2, mpt, ept // 0/ 12 $
$
 vecplot ugv, bgpdt, eqexin, cstm, casecc,/ugvb/0/0/1 $
 vecplot pgg, bgpdt, eqexin, cstm, casecc,/pggb/0/0/1 $
 vecplot qg, bgpdt, eqexin, cstm, casecc,/qgb/0/0/1 $
$
 sdr2   casecc, cstm, mpt, dit, eqexin,, ett, edt, bgpdt,pggb,
      qgb, ugvb, est, xycdb / opgb1, oqgb1, ougvb1,,,
      /'statics'/s,n,nosort2/v,y,nocomp=1 $
$
 sdr3   opgb1, oqgb1, ougvb1,,, / opgb2, oqgb2, ougvb2,,, / $
 sdr3   opg1, oqg1, ougv1,,, / opg2, oqg2, ougv2,,, / $
 sdr3   oes1, oef1, ,,, / oes2, oef2, ,,, / $
$
 output2   oef2, , , , // 0/12///'oef' $
 output2   oes2, , , , // 0/12///'oes' $
 output2   ougvb2, , , , // 0/12///'ougv2' $
 output2   opgb2, , , , // 0/12///'opg' $
 output2   oqgb2, , , , // 0/12///'oqg' $
 output2  ougv2, , , , // 0 / 12 / $
 output2   opg2, , , , // 0/12///'opg' $
 output2   oqg2, , , , // 0/12///'oqg' $
$
alter 189
$
 output2 ogpfb1, , , , // 0 / 12 / $
 output2 onrgy1  , , , , // 0/12///'onrgy' $
$
alter 197
$
 output2 oelof1, oelop1, , ,  // 0 / 12 / $
$
alter 210
$
 sdr3   ostr1,,,,, / ostr2,,,,, / $
 output2 ostr2  , , , , // 0/12///'oes' $
$
alter 219
$
 output2 ,,,,, // -9 / 12 / $

$************ end of msc_v675_sol24.alt ********************
$
```

APPENDIX B - ALTER for Solution 101

```
$
$*************************************************
$
$         msc_v675_sol101.alt
$
$ this dmap runs in msc/nastran solution 101 it writes the necessary datasets for norpost  processing.
$ basically, nastran version 67.5  is run and the output2 file is written in  compressed ascii format.  the
$ results are then transmitted to the ibm via ftp.  next a conversion program is run to convert the
$ compressed ascii formatted file to ibm binary.  this file is theninput to norpost.  the output database from
$ this run can be directly input to ncasa.
$
$    for the nastran run an assign card must be added
$    above the 'id' card.
$
$    assign output2='ncasa.op2', status=new,
$         form=formatted, unit=12
$
$ NOTE: If running ncasa on the workstation, use
$         form=unformatted
$*************************************************
$
compile subdmap = sedrcvr, souin = mscsou, nolist, noref $
alter 2,2
alter 44
$
compile subdmap = sestatic, souin = mscsou, nolist, noref $
$
alter 29
$
call johnout2  casedr, cstms, gpls, bgpdts, gpdts, uset,
           eqexins, geom2, mpt, ept,
           oug1, opg1, oqg1, oef1, oes1, oee1,
           ogps1, oesm1, oesg1, oeem1, oeeg1,
           onrgy, ogpfb / johna
$
compile subdmap = johnout2, list, noref $
```

APPENDIX B - ALTER for Solution 101 (Concluded)

```
$
$*********************************************************
$
subdmap johnout2  casedr, cstms, gpls, bgpdts, gpdts, uset,
           eqexins, geom2, mpt, ept,
            oug1, opg1, oqg1, oef1, oes1, oee1,
            ogps1, oesm1, oesg1, oeem1, oeeg1,
           onrgy, ogpfb / johna
$
 equivx casedr/casecc/always $
 equivx cstms/cstm/always $
 equivx gpls/gpl/always $
 equivx bgpdts/bgpdt/always $
 equivx gpdts/gpdt/always $
 equivx eqexins/eqexin/always $
 equivx ogpfb/ogpfb1/always $
$
   output2 ,,,,, //-1 / 12 /'statics'/ $
   output2 casecc, , , , // 0 / 12 / $
   output2 cstm, gpl, bgpdt, gpl, gpdt // 0 / 12 / $
   output2 uset, eqexin ,geom2, mpt, ept // 0/ 12 $
$
 sdr3   opg1, oqg1, oug1,oee1,, / opg2, oqg2, ougv2, ostr2,, / $
 sdr3   oes1, oef1, ,,, / oes2, oef2, ,,, / $
$
   output2 oef2, oes2,ogpfb1,onrgy , // 0 / 12 / $
   output2  ougv2, opg2, oqg2,  ostr2 // 0 / 12 / $
   output2 ,,,, // -9 / 12 / $
$    output2 opg1, , , , // 0/12///'matrix' $
$
return $
end $ johnout
$
$*************** end of subdmap johnout2 ********************
```